

QPA

Quivers and Path Algebras

Version 1.1

September 2010

The QPA-team

<http://sourceforge.net/projects/quiverspathalg/>
(See also <http://www.math.ntnu.no/~oyvinso/QPA/>)

We can add a comment here.

Address: Virginia Tech, Blacksburg, USA
NTNU, Trondheim, Norway

Abstract

Copyright

© 2010-2020 The QPA-team.

Acknowledgements

The system design of QPA was initiated by Edward L. Green, Lenwood S. Heath, and Craig A. Struble. It was continued and completed by Randall Cone and Edward Green. We would like to thank the following people for their contributions:

Gerard Brunick	Quivers, path algebras
Randall Cone	Code modernization and cleanup, GBNP interface (for Groebner bases), projective resolutions, user documentation
George Yuhasz	User documentation, matrix representations of path algebras

Colophon

This is the Colophon page.

Contents

1	Quick Start	6
1.1	Example 1	6
1.2	Example 2	7
1.3	Example 3	8
1.4	Example 4	8
2	Quivers	10
2.1	Information class, Quivers	10
2.1.1	InfoQuiver	10
2.2	Constructing Quivers	10
2.2.1	Quiver	10
2.2.2	OrderedBy	11
2.3	Categories and Properties of Quivers	11
2.3.1	IsQuiver	11
2.3.2	IsAcyclic	11
2.3.3	IsFinite	12
2.4	Orderings of paths in a quiver	12
2.5	Attributes and Operations for Quivers	12
2.5.1	12
2.5.2	VerticesOfQuiver	12
2.5.3	ArrowsOfQuiver	12
2.5.4	AdjacencyMatrixOfQuiver	13
2.5.5	GeneratorsOfQuiver	13
2.5.6	OrderOfQuiver	13
2.5.7	SizeOfQuiver	13
2.5.8	OrderingOfQuiver	13
2.6	Categories and Properties of Paths	14
2.6.1	IsPath	14
2.6.2	IsVertex	14
2.6.3	IsArrow	14
2.6.4	IsZeroPath	14
2.7	Attributes and Operations of Paths	14
2.7.1	SourceOfPath	14
2.7.2	TargetOfPath	15
2.7.3	LengthOfPath	15
2.7.4	WalkOfPath	15

2.7.5	*	15
2.7.6	=	16
2.7.7	<	16
2.8	Attributes of Vertices	16
2.8.1	IncomingArrowsOfVertex	16
2.8.2	OutgoingArrowsOfVertex	16
2.8.3	InDegreeOfVertex	16
2.8.4	OutDegreeOfVertex	16
2.8.5	NeighborsOfVertex	17
3	Path Algebras	18
3.1	Introduction	18
3.1.1	InfoPathAlgebra	18
3.2	Constructing Path Algebras	18
3.2.1	PathAlgebra	18
3.2.2	OrderedBy	18
3.3	Categories and Properties of Path Algebras	19
3.3.1	IsPathAlgebra	19
3.4	Attributes and Operations for Path Algebras	19
3.4.1	QuiverOfPathAlgebra	19
3.4.2		19
3.5	Operations on Path Algebra Elements	19
3.5.1	<	19
3.5.2	LeadingTerm	20
3.5.3	LeadingCoefficient	20
3.5.4	LeadingMonomial	20
3.5.5	IsLeftUniform	20
3.5.6	IsRightUniform	20
3.5.7	IsUniform	21
3.5.8	MappedExpression	21
3.6	Constructing Quotients of Path Algebras	21
3.7	Attributes of Ideals	21
3.8	Categories and Properties of Quotients of Path Algebras	22
3.8.1	IsSubAlgebraFpPathAlgebra	22
3.9	Attributes and Operations for Quotients of Path Algebras	22
3.9.1	NormalFormFunction	22
3.9.2	IsElementOfFpPathAlgebra	22
3.9.3	Coefficients	23
3.10	Attributes and Operations on Elements of Quotients of Path Algebra	23
3.10.1	IsNormalForm	23
3.10.2	<	23
3.10.3	ElementOfFpPathAlgebra	23
3.11	Predefined classes of quotient of path algebras	23
3.11.1	NakayamaAlgebra	23

4	Groebner Basis	24
4.1	Constructing a Groebner Basis	24
4.1.1	InfoGroebnerBasis	24
4.1.2	GroebnerBasis	24
4.2	Categories and Properties of Groebner Basis	24
4.2.1	IsGroebnerBasis	24
4.2.2	IsTipReducedGroebnerBasis	25
4.2.3	IsCompletelyReducedGroebnerBasis	25
4.2.4	IsHomogenousGroebnerBasis	25
4.2.5	IsCompleteGroebnerBasis	25
4.3	Attributes and Operations for Groebner Basis	25
4.3.1	CompletelyReduce	25
4.3.2	CompletelyReduceGroebnerBasis	26
4.3.3	TipReduce	26
4.3.4	TipReduceGroebnerBasis	26
4.3.5	Iterator	26
4.3.6	Enumerator	26
4.3.7	Nontips	27
4.3.8	AdmitsFinitelyManyNontips	27
4.3.9	NontipSize	27
4.3.10	IsPrefixOfTipInTipIdeal	27
4.4	Right Groebner Basis	28
4.4.1	IsRightGroebnerBasis	28
4.4.2	RightGroebnerBasisOfIdeal	28
4.4.3	RightGroebnerBasis	28
5	Using GBNP with Gap	29
5.1	GBNP	29
5.2	Setting up GBNP	29
5.3	Relevant GBNP internals	29
5.4	Communicating with GBNP	29
6	Right Modules over Path Algebras	30
6.1	Matrix Modules	30
6.1.1	RightModuleOverPathAlgebra	30
6.2	Categories Of Matrix Modules	32
6.2.1	IsPathAlgebraMatModule	32
6.3	Acting on Module Elements	32
6.3.1	\wedge	32
6.4	Vertex Projective Presentations	32
6.4.1	RightProjectiveModule	33
6.4.2	Vectorize	33
6.4.3	\wedge	34
6.4.4	$<$	34
6.4.5	$/$	34
A	An Appendix	35

Chapter 1

Quick Start

This chapter is intended for those who would like to get started with QPA right away by playing with a few examples. A simple example is presented first:

1.1 Example 1

We construct a quiver q , i.e. a finite directed graph, with one vertex and two loops:

```
gap> q := Quiver(["u"],[["u","u","a"],["u","u","b"]]);  
<quiver with 1 vertices and 2 arrows>
```

We can request the list of vertices and the list of arrows for q :

```
gap> VerticesOfQuiver(q);  
[ u ]  
gap> ArrowsOfQuiver(q);  
[ a, b ]
```

Next we create the path algebra pa from q over the rational numbers:

```
gap> pa := PathAlgebra(Rationals,q);  
<algebra-with-one over Rationals, with 3 generators>
```

In this case it is interesting to note that we've created an algebra isomorphic to the free algebra on two generators. We now retrieve and label the generators and multiplicative identity for pa :

```
gap> gens := GeneratorsOfAlgebra(pa);  
[ (1)*u, (1)*a, (1)*b ]  
gap> u := gens[1];  
(1)*u  
gap> a := gens[2];  
(1)*a  
gap> b := gens[3];  
(1)*b  
gap> id := One(pa);  
(1)*u
```

As we expect, in this case, the multiplicative identity for pa and the single vertex u are one in the same:

Example

```
gap> u = id;
true
```

We now create a list of generators for an ideal and ask for its Groebner basis:

Example

```
gap> polys := [a*b*a-b, b*a*b-b];
[ (-1)*b+(1)*a*b*a, (-1)*b+(1)*b*a*b ]
gap> gb := GBNPGroebnerBasis(polys, pa);
[ (-1)*a*b+(1)*b*a, (-1)*a*b+(1)*b^2, (-1)*b+(1)*a^2*b ]
```

Next, we create an ideal I in $\{\backslash\text{GAP}\}$ using the Groebner basis gb found above, and then the quotient pa/I :

Example

```
gap> I := Ideal(pa, gb);
<two-sided ideal in <algebra-with-one over Rationals, with 3
generators>,
(3 generators)>
```

Once we have the generators for a Groebner basis, we set the appropriate property for the ideal I :

Example

```
gap> grb := GroebnerBasis(I, gb);
<partial two-sided Groebner basis containing 3 elements>
```

1.2 Example 2

In this next example we create another path algebra that is essentially the free algebra on six generators. We then find the Groebner basis for a commutative example from (create bibliographic reference here) the book "Some Tapes of Computer Algebra" by A.M. Cohen, H. Cuyper, H. Sterk. We create the underlying quiver, and from it the path algebra over the rational numbers:

Example

```
gap> q := Quiver(["u"], [{"u", "u", "a"}, {"u", "u", "b"}, {"u", "u", "c"},
> [{"u", "u", "d"}, {"u", "u", "e"}, {"u", "u", "f"}]);
<quiver with 1 vertices and 6 arrows>
gap> fq := PathAlgebra(Rationals, q);
<algebra-with-one over Rationals, with 7 generators>
```

Next, the generators are labeled and the list of polynomials is entered:

Example

```
gap> gens := GeneratorsOfAlgebra(fq);
[ (1)*u, (1)*a, (1)*b, (1)*c, (1)*d, (1)*e, (1)*f ]
gap> u := gens[1]; a := gens[2]; b := gens[3]; c := gens[4];
gap> d := gens[5]; e := gens[6]; f := gens[7];
gap> polys := [ e*a,
> a^3 + f*a,
> a^9 + c*a^3,
> a^81 + c*a^9 + d*a^3,
> a^27 + d*a^81 + e*a^9 + f*a^3,
```

```

>      b + c*a^27 + e*a^81 + f*a^9,
>      c*b + d*a^27 + f*a^81,
>      a + d*b + e*a^27,
>      c*a + e*b + f*a^27,
>      d*a + f*b,
>      b^3 - b,
>      a*b - b*a, a*c - c*a,
>      a*d - d*a, a*e - e*a,
>      a*f - f*a, b*c - c*b,
>      b*d - d*b, b*e - e*b,
>      b*f - f*b, c*d - d*c,
>      c*e - e*c, c*f - f*c,
>      d*e - e*d, d*f - f*d,
>      e*f - f*e
> ;;

```

Finally, the Groebner basis is found:

Example

```

gap> gb := GBNPGroebnerBasis(polys,fq);
[ (1)*a, (1)*b, (-1)*c*d+(1)*d*c, (-1)*c*e+(1)*e*c, (-1)*d*e+(1)*e*d,
  (-1)*c*f+(1)*f*c, (-1)*d*f+(1)*f*d, (-1)*e*f+(1)*f*e ]

```

1.3 Example 3

The next example is from B. Keller's PhD thesis, p. 26:

Example

```

gap> q := Quiver(["u", "v"], [{"u", "v", "c"}, {"u", "u", "b"}, {"u", "u", "a"}]);
<quiver with 2 vertices and 3 arrows>
gap> pa := PathAlgebra(Rationals,q);
<algebra-with-one over Rationals, with 5 generators>
gap>
gap> # Get generators of path algebra:
gap> gens := GeneratorsOfAlgebra(pa);
[ (1)*u, (1)*v, (1)*c, (1)*b, (1)*a ]
gap> u := gens[1];; v := gens[2];; c := gens[3];;
gap> b := gens[4];; a := gens[5];; id := One(pa);;
gap>
gap> polys := [a*b*c+b*a*b+a*c];
[ (1)*c+(1)*a+(1)*b*a*b+(1)*a*b*c ]
gap> gb := GBNPGroebnerBasis(polys,pa);
[ (-1)*b*c+(1)*a*c, (1)*a+(1)*b*a*b, (1)*c+(1)*a*b*c, (-1)*b*a^2+(1)*a^2*b ]

```

1.4 Example 4

Here's an example that doesn't meet our necessary criteria that all elements in a generating set have monomials in the arrow ideal. Since the given path algebra is isomorphic to a free algebra, the single vertex is sent to the identity and there are no complications. First, we set up the algebra and generating set:

Example

```
gap> q := Quiver(["u"],[["u","u","x"],["u","u","y"]]);
<quiver with 1 vertices and 2 arrows>
gap> f := Rationals;
Rationals
gap> fq := PathAlgebra(f,q);
<algebra-with-one over Rationals, with 3 generators>
gap>
gap> # Get generators of path algebra:
gap> gens := GeneratorsOfAlgebra(fq);
[ (1)*u, (1)*x, (1)*y ]
gap> u := gens[1];; x := gens[2];; y := gens[3];; id := One(fq);;
gap> polys := [x*y-y*x,x^2*y-id,x*y^2-id];
[ (1)*x*y+(-1)*y*x, (-1)*u+(1)*x^2*y, (-1)*u+(1)*x*y^2 ]
```

Then we ask GBNP for its Groebner basis:

Example

```
gap> gb := GBNPGroebnerBasisNC(polys,fq);
The given path algebra is isomorphic to a free algebra.
[ (-1)*x+(1)*y, (-1)*u+(1)*x^3 ]
```

NOTE: It is important to realize that we've used the routine 'GBNPGroebnerBasisNC' which doesn't check that all elements in a given list have non-vertex monomials. So, if we run the standard QPA Groebner basis routine on this example, we get the following:

Example

```
gap> GBNPGroebnerBasis(polys,pa);
Please make sure all elements are in the given path algebra,
and each summand of each element is not (only) a vertex.
false
```

Chapter 2

Quivers

2.1 Information class, Quivers

A quiver Q is a set derived from a labeled directed multigraph with loops Γ . An element of Q is called a *path*, and falls into one of three classes. The first class is the set of *vertices* of Γ . The second class is the set of *walks* in Γ of length at least one, each of which is represented by the corresponding sequence of *arrows* in Γ . The third class is the singleton set containing the distinguished *zero path*, usually denoted 0. An associative multiplication is defined on Q .

This chapter describes the functions in QPA that deal with paths and quivers. The functions for constructing paths in Section 3.2 are normally not useful in isolation; typically, they are invoked by the functions for constructing quivers in Section 2.2.

2.1.1 InfoQuiver

◇ InfoQuiver

(info class)

is the info class for functions dealing with quivers.

2.2 Constructing Quivers

2.2.1 Quiver

◇ Quiver($N[, \text{arrow1}, \text{arrow2}, \dots]$)

(function)

◇ Quiver($[\text{vertex1}, \text{vertex2}, \dots,][\text{arrow1}, \text{arrow2}, \dots]$)

(function)

◇ Quiver(adjacencymatrix)

(function)

Returns: a quiver, which satisfies the property IsQuiver (2.3.1).

The first construction takes the number N of vertices and a list of arrows to specify the graph Γ and hence the quiver. The second construction takes a list of vertex names and a list of arrows for the same purpose. The third construction takes an adjacency matrix for the graph Γ .

In the first and third constructions, the vertices are named ‘v1, v2, ...’. In the second construction, unique vertex names are given as strings in the list that is the first parameter. Each arrow is a list consisting of a source vertex and a target vertex, followed optionally by an arrow name as a string.

Vertices and arrows are referenced as record components using the dot (‘.’) operator.

Example

```

gap> q1 := Quiver(["u","v"],[["u","u","a"],["u","v","b"],
> ["v","u","c"],["v","v","d"]]);
<quiver with 2 vertices and 4 arrows>
gap> VerticesOfQuiver(q1);
[ u, v ]
gap> ArrowsOfQuiver(q1);
[ a, b, c, d ]
gap> q2 := Quiver(2,[1,1],[2,1],[1,2]);
<quiver with 2 vertices and 3 arrows>
gap> ArrowsOfQuiver(q2);
[ a1, a2, a3 ]
gap> VerticesOfQuiver(q2);
[ v1, v2 ]
gap> q3 := Quiver(2,[1,1,"a"],[2,1,"b"],[1,2,"c"]);
<quiver with 2 vertices and 3 arrows>
gap> ArrowsOfQuiver(q3);
[ a, b, c ]
gap> q4 := Quiver([1,1],[2,1]);
<quiver with 2 vertices and 5 arrows>
gap> VerticesOfQuiver(q4);
[ v1, v2 ]
gap> ArrowsOfQuiver(q4);
[ a1, a2, a3, a4, a5 ]
gap> SourceOfPath(q4.a2);
v1
gap> TargetOfPath(q4.a2);
v2

```

2.2.2 OrderedBy

◇ `OrderedBy(quiver, ordering)` (function)

Returns: a copy of *quiver* whose elements are ordered by *ordering*. The default ordering of a quiver is length left lexicographic. See Section 2.4 for more information.

2.3 Categories and Properties of Quivers

2.3.1 IsQuiver

◇ `IsQuiver(object)` (property)

is true when *object* is a quiver.

2.3.2 IsAcyclic

◇ `IsAcyclic(object)` (property)

is true when *object* is a quiver with no cycles.

2.3.3 IsFinite

◇ `IsFinite(object)`

(property)

is true when *object* is a finite set. Synonymous with ‘IsAcyclic’.

Example

```
gap> quiver1 := Quiver(2, [[1,2]]);
<quiver with 2 vertices and 1 arrows>
gap> IsQuiver("v1");
false
gap> IsQuiver(quiver1);
true
gap> IsAcyclic(quiver1);
true
gap> quiver2 := Quiver(["u", "v"], [{"u", "v"}, {"v", "u"}]);
<quiver with 2 vertices and 2 arrows>
gap> IsAcyclic(quiver2);
false
gap> IsFinite(quiver1);
true
gap> IsFinite(quiver2);
false
```

2.4 Orderings of paths in a quiver

To be written.

2.5 Attributes and Operations for Quivers

2.5.1 .

◇ `.(Q, element)`

(operation)

The operation `.` operates on Q , a quiver, and an element, a vertex or an arrow, to allow access to generators of the quiver. If you have named your vertices and arrows the the access looks like ‘ $Q.name\ of\ element$ ’. If you have not named the elements of the quiver then the default names are v_1, v_2, \dots and a_1, a_2, \dots in the order they are created.

2.5.2 VerticesOfQuiver

◇ `VerticesOfQuiver(object)`

(attribute)

An attribute. Returns a list of paths that are vertices in *object*.

2.5.3 ArrowsOfQuiver

◇ `ArrowsOfQuiver(object)`

(attribute)

An attribute. Returns a list of paths that are arrows in *object*.

2.5.4 AdjacencyMatrixOfQuiver

◇ AdjacencyMatrixOfQuiver(*object*) (attribute)

An attribute. Returns the adjacency matrix of *object*.

2.5.5 GeneratorsOfQuiver

◇ GeneratorsOfQuiver(*object*) (attribute)

An attribute. Returns a list of the vertices and the arrows in *object*.

2.5.6 OrderOfQuiver

◇ OrderOfQuiver(*object*) (attribute)

An attribute. Returns the number of vertices in *object*.

2.5.7 SizeOfQuiver

◇ SizeOfQuiver(*object*) (attribute)

An attribute. Returns the number of arrows in *object*.

2.5.8 OrderingOfQuiver

◇ OrderingOfQuiver(*object*) (attribute)

An attribute. Returns the ordering used to order elements in *object*. See Section 2.4 for more information.

Example

```
gap> q1 := Quiver(["u", "v"], [{"u", "u", "a"}, {"u", "v", "b"},
> ["v", "u", "c"}, {"v", "v", "d"}]);
<quiver with 2 vertices and 4 arrows>
gap> q1.a;
a
gap> q1.v;
v
gap> VerticesOfQuiver(q1);
[ u, v ]
gap> ArrowsOfQuiver(q1);
[ a, b, c, d ]
gap> AdjacencyMatrixOfQuiver(q1);
[ [ 1, 1 ], [ 1, 1 ] ]
gap> GeneratorsOfQuiver(q1);
[ u, v, a, b, c, d ]
gap> OrderOfQuiver(q1);
2
gap> SizeOfQuiver(q1);
4
```

```
gap> OrderingOfQuiver(q1);
<length left lexicographic ordering>
```

2.6 Categories and Properties of Paths

2.6.1 IsPath

◇ `IsPath(object)` (category)

All path objects are in this category.

2.6.2 IsVertex

◇ `IsVertex(object)` (category)

All vertices are in this category.

2.6.3 IsArrow

◇ `IsArrow(object)` (category)

All arrows are in this category.

2.6.4 IsZeroPath

◇ `IsZeroPath(object)` (property)

is true when *object* is the zero path.

Example

```
gap> q1 := Quiver(["u", "v"], [{"u", "u", "a"}, {"u", "v", "b"},
> ["v", "u", "c"}, {"v", "v", "d"}]);
<quiver with 2 vertices and 4 arrows>
gap> IsPath(q1.b);
true
gap> IsPath(q1.u);
true
gap> IsVertex(q1.c);
false
gap> IsZeroPath(q1.d);
false
```

2.7 Attributes and Operations of Paths

2.7.1 SourceOfPath

◇ `SourceOfPath(object)` (attribute)

An attribute. Returns the source (first) vertex of *object*.

2.7.2 TargetOfPath

◇ TargetOfPath(*object*)

(attribute)

An attribute. Returns the target (last) vertex of *object*.

2.7.3 LengthOfPath

◇ LengthOfPath(*object*)

(attribute)

An attribute. Returns the length of *object*.

2.7.4 WalkOfPath

◇ WalkOfPath(*object*)

(attribute)

An attribute. Returns a list of the arrows that constitute *object* in order.

2.7.5 *

◇ *(*p*, *q*)

(operation)

The operation $*$ operates on p and q , which are two paths in the same quiver. It returns the multiplication of the paths. If the paths are not in the same quiver an error is returned. If the target of p differs from the source of q , then the result is the zero path. Otherwise, if either path is a vertex, then the result is the other path. Finally, if both are paths of length at least 1, then the result is the concatenation of the walks of the two paths.

Example

```
gap> q1 := Quiver(["u", "v"], [{"u", "u", "a"}, {"u", "v", "b"},
> ["v", "u", "c"}, {"v", "v", "d"}]);
<quiver with 2 vertices and 4 arrows>
gap> SourceOfPath(q1.v);
v
gap> p1:=q1.a*q1.b*q1.d*q1.d;
a*b*d^2
gap> TargetOfPath(p1);
v
gap> p2:=q1.b*q1.b;
0
gap> WalkOfPath(p1);
[ a, b, d, d ]
gap> WalkOfPath(q1.a);
[ a ]
gap> LengthOfPath(p1);
4
gap> LengthOfPath(q1.v);
0
```

2.7.6 =

$\diamond = (p, q)$ (operation)

The operation `=` operates on two paths p and q , and compares the paths for equality. Two paths are equal if they have the same source and the same target and if they have the same walks. The paths p and q must be in the same quiver.

2.7.7 <

$\diamond < (p, q)$ (operation)

The operation `<` operates on two paths p and q , and compares them with respect to the ordering of the quiver. This is meaningful only if p and q are in the same quiver.

Example

```
gap> q1.a=q1.b;
false
gap> q1.a < q1.v;
false
gap> q1.a < q1.c;
true
```

2.8 Attributes of Vertices

2.8.1 IncomingArrowsOfVertex

$\diamond \text{IncomingArrowsOfVertex}(\text{object})$ (attribute)

An attribute. Returns a list of arrows having *object* as target. Only meaningful if *object* is in a quiver.

2.8.2 OutgoingArrowsOfVertex

$\diamond \text{OutgoingArrowsOfVertex}(\text{object})$ (attribute)

An attribute. Returns a list of arrows having *object* as source.

2.8.3 InDegreeOfVertex

$\diamond \text{InDegreeOfVertex}(\text{object})$ (attribute)

An attribute. Returns the number of arrows having *object* as target. Only meaningful if *object* is in a quiver.

2.8.4 OutDegreeOfVertex

$\diamond \text{OutDegreeOfVertex}(\text{object})$ (attribute)

An attribute. Returns the number of arrows having *object* as source.

2.8.5 NeighborsOfVertex

◇ `NeighborsOfVertex(object)`

(attribute)

An attribute. Returns a list of neighbors of *object*, that is, vertices that are targets of arrows having *object* as source.

Example

```
gap> q1 := Quiver(["u", "v"], [{"u", "u", "a"}, {"u", "v", "b"},  
> ["v", "u", "c"}, {"v", "v", "d"}]);  
<quiver with 2 vertices and 4 arrows>  
gap> OutgoingArrowsOfVertex(q1.u);  
[ a, b ]  
gap> InDegreeOfVertex(q1.u);  
2  
gap> NeighborsOfVertex(q1.v);  
[ u, v ]
```

Chapter 3

Path Algebras

3.1 Introduction

A path algebra is an algebra constructed from a field F and a quiver Q . The path algebra FQ contains all finite linear combinations of elements of Q . This chapter describes the functions in QPA that deal with path algebras and quotients of path algebras. Path algebras are algebras so see Chapter "ref:algebras" for functionality such as generators, basis functions, and mappings.

3.1.1 InfoPathAlgebra

◇ InfoPathAlgebra

(info class)

is the info class for functions dealing with path algebras.

3.2 Constructing Path Algebras

3.2.1 PathAlgebra

◇ PathAlgebra(F , Q)

(function)

Returns: the path algebra FG of Q over the field F .

Example

```
gap> q := Quiver(["u", "v"], [{"u", "u", "a"}, {"u", "v", "b"},  
> ["v", "u", "c"}, {"v", "v", "d"}]);  
<quiver with 2 vertices and 4 arrows>  
gap> f := FiniteField(23);  
GF(23)  
gap> fq := PathAlgebra(f, q);  
<algebra-with-one over GF(23), with 6 generators>
```

* NO: the FOLLOWING is not true:

3.2.2 OrderedBy

◇ OrderedBy($path$, $algebra$, $ordering$)

(function)

Returns: a copy of $path$ $algebra$ whose elements are ordered by $ordering$. See Section 2.4 for more information on orderings.

3.3 Categories and Properties of Path Algebras

3.3.1 IsPathAlgebra

◇ `IsPathAlgebra(object)`

(property)

is true when *object* is a path algebra.

Example

```
gap> IsPathAlgebra(fq);
true
gap> IsPathAlgebra(q);
false
```

3.4 Attributes and Operations for Path Algebras

3.4.1 QuiverOfPathAlgebra

◇ `QuiverOfPathAlgebra(object)`

(attribute)

An attribute. Returns the quiver from which *object* was constructed.

Example

```
gap> QuiverOfPathAlgebra(fq);
<quiver with 2 vertices and 4 arrows>
```

3.4.2 .

◇ `.(FQ, element)`

(operation)

The operation `.` operates on a path algebra FQ and an element *element*, which is a vertex or an arrow in the quiver Q . It returns the generator as an element of the path algebra.

Example

```
gap> fq.v;
(Z(23)^0)*v
gap> fq.b;
(Z(23)^0)*b
```

`>OrderingOfAlgebra(path algebra)` returns the ordering of the quiver of the path algebra.

3.5 Operations on Path Algebra Elements

3.5.1 <

◇ `<(a, b)`

(operation)

The operation `<` operates on elements *a* and *b* of a path algebra FQ , and they are compared using the ordering for the path algebra. See Section 2.4 for more information on orderings.

3.5.2 LeadingTerm

◇ `LeadingTerm(element)`

(operation)

◇ `Tip(element)`

(operation)

The operation `LeadingTerm` or equivalently `Tip` operates on an x in a path algebra, and it returns the term in *element* whose monomial is largest among those monomials that have nonzero coefficients; this term is known as the **tip** of *element*.

3.5.3 LeadingCoefficient

◇ `LeadingCoefficient(element)`

(operation)

◇ `TipCoefficient(element)`

(operation)

The operation `LeadingCoefficient` or equivalently `TipCoefficient` operates on an *element* in a path algebra, and it returns the coefficient of the tip of *element*. This is an element of the field.

3.5.4 LeadingMonomial

◇ `LeadingMonomial(element)`

(operation)

◇ `TipMonomial(element)`

(operation)

The operation `LeadingMonomial` or equivalently `TipMonomial` operates on an *element* in a path algebra, and it returns the monomial of the tip of *element*; it is the largest monomial occurring in *element* with a nonzero coefficient. This is an element of the underlying quiver, not of the path algebra.

Example

```
gap> elem := fq.a*fq.b*fq.c + fq.b*fq.d*fq.c+fq.d*fq.d;
(Z(23)^0)*d^2+(Z(23)^0)*a*b*c+(Z(23)^0)*b*d*c
gap> LeadingTerm(elem);
(Z(23)^0)*b*d*c
gap> LeadingCoefficient(elem);
Z(23)^0
gap> LeadingMonomial(elem);
b*d*c
```

3.5.5 IsLeftUniform

◇ `IsLeftUniform(element)`

(operation)

The operation `IsLeftUniform` operates on an *element* in a path algebra, and it returns true if each monomial in *element* has the same source vertex.

3.5.6 IsRightUniform

◇ `IsRightUniform(element)`

(operation)

The operation `IsRightUniform` operates on an *element* in a path algebra, and it returns true if each monomial in *element* has the same target vertex.

3.5.7 IsUniform

◇ `IsUniform(element)`

(operation)

The operation `IsUniform` operates on an *element* in a path algebra, and it returns true if each monomial in *element* has both the same source vertex and the same target vertex.

Example

```
gap> IsLeftUniform(elem);
false
gap> IsRightUniform(elem);
false
gap> IsUniform(elem);
false
gap> another := fq.a*fq.b + fq.b*fq.d*fq.c*fq.b*fq.d;
(Z(23)^0)*a*b+(Z(23)^0)*b*d*c*b*d
gap> IsLeftUniform(another);
true
gap> IsRightUniform(another);
true
gap> IsUniform(another);
true
```

3.5.8 MappedExpression

◇ `MappedExpression(expr, gens1, gens2)`

(operation)

The operation `MappedExpression` operates on *expr* from a path algebra and two equal-length lists of generators *gens1* and *gens2* for subalgebras. The *expr* must be in the subalgebra generated by *gens1*. The lists define a mapping of each generator in *gens1* to the corresponding generator in *gens2*. The value returned is the evaluation of the mapping at *expr*.

3.6 Constructing Quotients of Path Algebras

See Chapter "ref:algebras" on how to construct an ideal and a quotient of an algebra. When the quotient is constructed, it is still a path algebra and thus the same commands may be used with quotients. Also since a quotient is still an algebra, refer to "ref:algebras".

Example

```
gap> I := Ideal(fq, [fq.a * fq.b, fq.d * fq.d - fq.b * fq.c]);
<two-sided ideal in <algebra-with-one over GF(23), with 6 generators>,
(2 generators)>
gap> GeneratorsOfIdeal(I);
[ (Z(23)^0)*a*b, (Z(23)^11)*b*c+(Z(23)^0)*d^2 ]
gap> quot := fq/I;
<algebra-with-one over GF(23), with 6 generators>
```

3.7 Attributes of Ideals

Groebner Basis Of an Ideal: For many of the functions related to quotients, you will need to compute a Groebner basis of the ideal. Refer to the chapters "qpa:groebner basis" and "qpa:using opal with

gap” to learn more.

Example

```
gap> J := Ideal(fq, [fq.a*fq.b]);
<two-sided ideal in <algebra-with-one over GF(23), with 6 generators>,
  (1 generators)>
gap> anotherquot := fq/J;
<algebra-with-one over GF(23), with 6 generators>
gap> gb := GroebnerBasis(J, [fq.a*fq.b]);
<complete two-sided Groebner basis containing 1 elements>
gap> SetIsCompleteGroebnerBasis(gb, true);
gap> IsCompleteGroebnerBasis(gb);
true
gap> gb = GroebnerBasisOfIdeal(J);
true
```

3.8 Categories and Properties of Quotients of Path Algebras

3.8.1 IsSubAlgebraFpPathAlgebra

◇ IsSubAlgebraFpPathAlgebra(*object*)

(property)

is true when *object* is a quotient of a path algebra.

Example

```
gap> IsSubalgebraFpPathAlgebra(quot);
true
gap> IsSubalgebraFpPathAlgebra(fq);
false
```

3.9 Attributes and Operations for Quotients of Path Algebras

3.9.1 NormalFormFunction

◇ NormalFormFunction(*object*)

(attribute)

is a function that can compute normal forms for elements of *object*. It may be supplied by the user.

3.9.2 IsElementOfFpPathAlgebra

◇ IsElementOfFpPathAlgebra(*object*)

(property)

is true if *object* is an element of some quotient of a path algebra.

Example

```
gap> this := anotherquot.a*anotherquot.b;
[(Z(23)^0)*a*b]
gap> IsElementOfFpPathAlgebra(this);
true
gap> IsElementOfFpPathAlgebra(fq);
false
```

3.9.3 Coefficients

◇ `Coefficients(element)`

(operation)

The operation `Coefficients` operates on an *element* of a quotient of a path algebra, and it returns the coefficients of the *element* in terms of its canonical basis. Question: Does this only take one argument?

3.10 Attributes and Operations on Elements of Quotients of Path Algebra

3.10.1 IsNormalForm

◇ `IsNormalForm(element)`

(operation)

The operation `IsNormalForm` operates on an *element* from a path algebra, and it is true if *element* is known to be in normal form.

Example

```
gap> IsNormalForm(this);
false
```

3.10.2 <

◇ `<(a, b)`

(operation)

The operation `<` operates on two elements *a* and *b* from a path algebra FQ , and it compares them using the ordering for the path algebra.

3.10.3 ElementOfFpPathAlgebra

◇ `ElementOfFpPathAlgebra(family, element, computenormal)`

(operation)

The operation `ElementOfFpPathAlgebra` operates on an *element* in a path algebra, and it projects it into the quotient given by *family*. If *computenormal* is true, then the normal form of the projection of *element* is returned.

3.11 Predefined classes of quotient of path algebras

3.11.1 NakayamaAlgebra

◇ `NakayamaAlgebra(admiss-seq, field)`

(function)

Returns: the Nakayama algebra corresponding to the admissible sequence *admiss-seq* over the field *field*, or the admissible sequence entered if entered sequence is not an admissible sequence.

This function creates a Nakayama algebra from an admissible sequence over a field.

Chapter 4

Groebner Basis

This chapter contains the declarations and implementations needed for Groebner basis. Currently, we do not provide algorithms to actually compute Groebner basis; instead, the declarations and implementations are provided here for GAP objects and the actual elements of Groebner basis are expected to be computed by external packages such as ‘Opal’ and ‘Groebner’.

4.1 Constructing a Groebner Basis

4.1.1 InfoGroebnerBasis

◇ InfoGroebnerBasis (info class)

is the info class for functions dealing with Groebner basis.

4.1.2 GroebnerBasis

◇ GroebnerBasis(A , $rels$) (function)

Returns: an object in the ‘IsGroebnerBasis’ category.

The ‘GroebnerBasis’ global function takes an algebra A and a list of relations $rels \in A$ and creates an object in the ‘IsGroebnerBasis’ category. There are absolutely no checks for correctness in this function. Giving a set of relations that does not form a Groebner basis may result in incorrect answers or unexpected errors. This function is intended to be used by packages providing access to external Groebner basis programs.

4.2 Categories and Properties of Groebner Basis

4.2.1 IsGroebnerBasis

◇ IsGroebnerBasis($object$) (property)

is true when $object$ is a Groebner basis and ‘false’ otherwise.

gap>

Example

4.2.2 IsTipReducedGroebnerBasis

◇ `IsTipReducedGroebnerBasis(object)`

(property)

is true when *object* is a Groebner basis which is tip reduced.

Example

```
gap>
```

4.2.3 IsCompletelyReducedGroebnerBasis

◇ `IsCompletelyReducedGroebnerBasis(object)`

(property)

is true when *object* is a Groebner basis which is completely reduced.

Example

```
gap>
```

4.2.4 IsHomogenousGroebnerBasis

◇ `IsHomogenousGroebnerBasis(object)`

(property)

is true when *object* is a Groebnerbasis which is homogenous.

Example

```
gap>
```

4.2.5 IsCompleteGroebnerBasis

◇ `IsCompleteGroebnerBasis(object)`

(property)

is true when *object* is a complete Groebner basis. While philosophically something that isn't a complete Groebner basis isn't a Groebner basis at all, this property can be used in conjunction with other properties to see if the the Groebner basis contains enough information for computations. An example of a system that creates incomplete Groebner basis is 'Opal'.

Example

```
gap>
```

4.3 Attributes and Operations for Groebner Basis

4.3.1 CompletelyReduce

◇ `CompletelyReduce(GB, a)`

(operation)

The operation `CompletelyReduce` operates on an element *a* in a path algebra, and it reduces *a* by the Groebner basis *GB*. The reduced element is returned. If *a* is already completely reduced, the original element *a* is returned.

Example

```
gap>
```

4.3.2 CompletelyReduceGroebnerBasis

◇ `CompletelyReduceGroebnerBasis(GB)`

(operation)

The operation `CompletelyReduceGroebnerBasis` operates on a Groebner basis GB , and it modifies a Groebner basis GB such that each relation in GB is completely reduced. The ‘`IsCompletelyReducedGroebnerBasis`’ and ‘`IsTipReducedGroebnerBasis`’ properties are set as a result of this operation. The resulting relations will be placed in sorted order according to the ordering of GB .

gap>

Example

4.3.3 TipReduce

◇ `TipReduce(GB, a)`

(operation)

The operation `TipReduce` operates on an element a in a path algebra, and it tip reduced by the Groebner basis GB . If a is already tip reduced, then the original a is returned.

gap>

Example

4.3.4 TipReduceGroebnerBasis

◇ `TipReduceGroebnerBasis(GB)`

(operation)

The operation `TipReduceGroebnerBasis` operates on Groebner basis GB , and it returns an equivalent Groebner basis to GB such that each relation generating GB is tip reduced. If GB is already tip reduced, this function returns the original object GB , possibly with the addition of the ‘`IsTipReduced`’ property set.

gap>

Example

4.3.5 Iterator

◇ `Iterator(GB)`

(operation)

The operation `Iterator` operates on a Groebner basis GB , and it creates an iterator that iterates over the relations making up the Groebner basis. These relations are iterated over in ascending order with respect to the ordering for the family the elements are contained in.

gap>

Example

4.3.6 Enumerator

◇ `Enumerator(GB)`

(operation)

The operation `Enumerator` operates on a Groebner basis GB , and it creates an enumerator that enumerates the relations making up the Groebner basis. These relations should be enumerated in ascending order with respect to the ordering for the family the elements are contained in.

Example

```
gap>
```

4.3.7 Nontips

◇ `Nontips(GB)` (operation)

The operation `Nontips` operates on a Groebner basis GB , and it returns a list of nontip elements for a Groebner basis. In order to compute the nontip elements, the Groebner basis must be complete and tip reduced, and there must be a finite number of nontips. If there are an infinite number of nontips, the operation returns ‘fail’.

Example

```
gap>
```

4.3.8 AdmitsFinitelyManyNontips

◇ `AdmitsFinitelyManyNontips(GB)` (operation)

The operation `AdmitsFinitelyManyNontips` operates on a Groebner basis GB , and it returns ‘true’ if the Groebner basis admits only finitely many nontips and ‘false’ otherwise. This operation only applies to complete Groebner basis.

Example

```
gap>
```

4.3.9 NontipSize

◇ `NontipSize(GB)` (operation)

The operation `NontipSize` operates on a Groebner basis GB , and it returns the number of nontips admitted by the Groebner basis GB . This operation is available only to complete basis.

Example

```
gap>
```

4.3.10 IsPrefixOfTipInTipIdeal

◇ `IsPrefixOfTipInTipIdeal(GB, R)` (operation)

The operation `IsPrefixOfTipInTipIdeal` operates on a Groebner basis GB and a relation R , and it checks the tip of a relation R to see if it is in the tip ideal generated by the tips of GB . This is used mainly for the construction of right Groebner basis, but is made available for general use in case there are other unforeseen applications.

Example

```
gap>
```

4.4 Right Groebner Basis

In this section we support right Groebner basis for two-sided ideals with Groebner basis. More general cases may be supported in the future.

4.4.1 IsRightGroebnerBasis

◇ `IsRightGroebnerBasis(object)` (property)

is true when *object* a right Groebner basis.

Example

```
gap>
```

4.4.2 RightGroebnerBasisOfIdeal

◇ `RightGroebnerBasisOfIdeal(object)` (attribute)

An attribute. Stores a right Groebner basis of a right ideal, *object*, is one has been computed.

4.4.3 RightGroebnerBasis

◇ `RightGroebnerBasis(R)` (operation)

The operation `RightGroebnerBasis` operates on a right ideal *R* in a path algebra, and it constructs a right Groebner basis for the right ideal, *R*, which must support a right Groebner basis theory. Right now, this requires that *R* have a complete Groebner basis.

Chapter 5

Using GBNP with Gap

5.1 GBNP

GNBP is a non-commutative Groebner Basis package which assumes that all algebras involved will be free algebras over a finite number of non-commuting generators. It also assumes that the ordering on the monomials is left length-lexicographic.

5.2 Setting up GBNP

in progress...

5.3 Relevant GBNP internals

in progress...

5.4 Communicating with GBNP

in progress...

Chapter 6

Right Modules over Path Algebras

There are two implementations of right modules over path algebras. The first type are matrix modules that are defined by vector spaces and linear transformations. The second type are presentations defined by vertex projective modules.

6.1 Matrix Modules

The first implementation of right modules over path algebras views them as a collection of vector spaces and linear transformations. Each vertex in the path algebra is associated with a vector space over the field of the algebra. For each vertex v of the algebra there is a vector space V . Arrows of the algebra are then associated with linear transformations which map the vector space of the source vertex to the vector space of the target vertex. For example, if a is an arrow from v to w then there is a transformation from vector space V to W . In practice when creating the modules all we need to know is the transformations and we can create the vector spaces of the correct dimension, and check to make sure the dimensions all agree. We can create a module in this way as follows.

6.1.1 RightModuleOverPathAlgebra

◇ `RightModuleOverPathAlgebra(A, mats)` (function)

Returns: a representation over a path algebra.

In the function call, A is the path algebra. The list of matrices $mats$ can take on three different forms.

1) The argument $mats$ can be a list of blocks of matrices where each block is of the form, “[”name of arrow”,matrix]”. So if you named your arrows when you created the quiver, then you can associate a matrix with that arrow explicitly.

2) The argument $mats$ is just a list of matrices, and the matrices will be associated to the arrows in the order of arrow creation. If when creating the quiver, the arrow a was created first, then a would be associated with the first matrix.

3) The method is very much the same as the second method. If $arrows$ is a list of the arrows of the quiver (obtained for instance through `arrows := ArrowsOfQuiver(Q);`), the argument $mats$ can have the format `[[arrows[1],matrix_1],[arrows[2],matrix_2],...]`.

If you would like the trivial vector space at any vertex, then for each incoming arrow “ a ”, associate it with a list of the form `["a", [n, 0]]` where n is the dimension of the vector space at the source vertex

of the arrow. Likewise for all outgoing arrows "b", associate them to a block of form $["b", [0, n]]$ where n is the dimension of the vector space at the target vertex of the arrow.

A warning though, the function assumes that you do not mix the styles of inputting the matrices/linear transformations associated to the arrows in the quiver. Furthermore, each arrow needs to be assigned a matrix, otherwise an error will be returned. The function verifies that the dimensions of the matrices and vector spaces are correct and match, and that each arrow has only one matrix assigned to it.

Example

```
gap> Q := Quiver(2, [[1, 2, "a"], [2, 1, "b"], [1, 1, "c"]]);
<quiver with 2 vertices and 3 arrows>
gap> P := PathAlgebra(Rationals, Q);
<algebra-with-one over Rationals, with 5 generators>
gap> matrices := [{"a", [[1,0,0],[0,1,0]]}, {"b", [[0,1],[1,0],[0,1]]},
> {"c", [[0,0],[1,0]]}];
[ [ "a", [ [ 1, 0, 0 ], [ 0, 1, 0 ] ] ],
  [ "b", [ [ 0, 1 ], [ 1, 0 ], [ 0, 1 ] ] ],
  [ "c", [ [ 0, 0 ], [ 1, 0 ] ] ] ]
gap> M := RightModuleOverPathAlgebra(P, matrices);
<right-module over <algebra-with-one over Rationals, with 5
generators>>
gap> mats := [ [[1,0,0], [0,1,0]], [[0,1],[1,0],[0,1]], [[0,0],[1,0]] ];
gap> N := RightModuleOverPathAlgebra(P, mats);
<right-module over <algebra-with-one over Rationals, with 5
generators>>
gap> arrows := ArrowsOfQuiver(Q);
[ a, b, c ]
gap> mats := [arrows[1], [[1,0,0],[0,1,0]],
> [arrows[2], [[0,1],[1,0],[0,1]], [arrows[3], [[0,0],[1,0]]]];
gap> N := RightModuleOverPathAlgebra(P, mats);
<right-module over <algebra-with-one over Rationals, with 5
generators>>
gap> # Next we give the vertex simple associate to vertex 1.
gap> M :=
RightModuleOverPathAlgebra(P, [{"a", [1,0]}, {"b", [0,1]}, {"c", [[0]]}]);
<right-module over <algebra-with-one over Rationals, with 5
generators>>
gap> # Finally, the next defines the zero representation of the quiver.
gap> M :=
RightModuleOverPathAlgebra(P, [{"a", [0,0]}, {"b", [0,0]}, {"c", [0,0]}]);
<right-module over <algebra-with-one over Rationals, with 5
generators>>
gap> Dimension(M);
0
gap> Basis(M);
Basis( <
0-dimensional right-module over <algebra-with-one over Rationals, with
5 generators>>, [ ] )
```

6.2 Categories Of Matrix Modules

6.2.1 IsPathAlgebraMatModule

◇ IsPathAlgebraMatModule(*object*)

(filter)

These matrix modules fall under the category ‘IsAlgebraModule’ with the added filter of ‘IsPathAlgebraMatModule’. Operations available for algebra modules can be applied to path algebra modules. See “ref:representations of algebras” for more details. These modules are also vector spaces over the field of the path algebra. So refer to “ref:vector spaces” for descriptions of the basis and elementwise operations available.

6.3 Acting on Module Elements

6.3.1 ^

◇ ^(*m*, *p*)

(operation)

The operation ^ operates on an element *m* in a module and a path *p* in a path algebra, and it returns the element *m* multiplied with *p*. When you act on an module element *m* by an arrow *a* from *v* to *w*, the component of *m* from *V* is acted on by *L* the transformation associated to *a* and placed in the component *W*. All other components are given the value 0.

Example

```
gap> # Using the path algebra P from the above example.
gap> matrices := [{"a", [[1,0,0],[0,1,0]]}, {"b", [[0,1],[1,0],[0,1]]},
> {"c", [[0,0],[1,0]]}];
gap> [ ["a", [ [ 1, 0, 0 ], [ 0, 1, 0 ] ] ],
      ["b", [ [ 0, 1 ], [ 1, 0 ], [ 0, 1 ] ] ],
      ["c", [ [ 0, 0 ], [ 1, 0 ] ] ] ]
gap> M := RightModuleOverPathAlgebra(P,matrices);
<right-module over <algebra-with-one over Rationals, with 5
generators>>
gap> B:=BasisVectors(Basis(M));
gap> [ [ [ 1, 0 ], [ 0, 0, 0 ] ], [ [ 0, 1 ], [ 0, 0, 0 ] ],
      [ [ 0, 0 ], [ 1, 0, 0 ] ], [ [ 0, 0 ], [ 0, 1, 0 ] ],
      [ [ 0, 0 ], [ 0, 0, 1 ] ] ]
gap> B[1]+B[3];
[ [ 1, 0 ], [ 1, 0, 0 ] ]
gap> 4*B[2];
[ [ 0, 4 ], [ 0, 0, 0 ] ]
gap> m:=5*B[1]+2*B[4]+B[5];
[ [ 5, 0 ], [ 0, 2, 1 ] ]
gap> m^(P.a*P.b-P.c);
[ [ 0, 5 ], [ 0, 0, 0 ] ]
gap> B[1]^P.a;
[ [ 0, 0 ], [ 1, 0, 0 ] ]
gap> B[2]^P.b;
[ [ 0, 0 ], [ 0, 0, 0 ] ]
gap> B[4]^(P.b*P.c);
[ [ 0, 0 ], [ 0, 0, 0 ] ]
```


6.4 Vertex Projective Presentations

In general, if R is a ring and e is an idempotent of R then eR is a projective module of R . Then we can form a direct sum of these projective modules together to form larger projective module. One can construct more general modules by providing a *vertex projective presentation*. In this case, M is the cokernel as given by the following exact sequence: $\bigoplus_{j=1}^r w(j)R \rightarrow \bigoplus_{i=1}^s v(i)R \rightarrow M \rightarrow 0$ for some map between $\bigoplus_{j=1}^r w(j)R$ and $\bigoplus_{i=1}^s v(i)R$. The maps w and v map the integers to some idempotent in R .

6.4.1 RightProjectiveModule

◇ **RightProjectiveModule**(A , $verts$) (function)

Returns: the right projective module over A which is the direct sum of projective modules of the form vA where the vertices are taken from the list of vertices $verts$.

In this implementation the algebra can be a quotient of a path algebra. So if the list was $[v, w]$ then the module created will be the direct sum $vA \oplus wA$, in that order. Elements of the modules are vectors of algebra elements, and in each component, each path begins with the vertex in that position in the list of vertices. Right projective modules are implemented as algebra modules (see "ref:Representations of Algebras") and all operations for algebra modules are applicable to right projective modules. In particular, one can construct submodules using 'SubAlgebraModule'.

Here we create the right projective module $P = vA \oplus vA \oplus wA$.

Example

```
gap> F:=GF(11);
GF(11)
gap> Q:=Quiver(["v", "w", "x"], [{"v", "w", "a"}, {"v", "w", "b"}, {"w", "x", "c"}]);
<quiver with 3 vertices and 3 arrows>
gap> A:=PathAlgebra(F, Q);
<algebra-with-one over GF(11), with 6 generators>
gap> P:=RightProjectiveModule(A, [A.v, A.v, A.w]);
<right-module over <algebra-with-one over GF(11), with 6 generators>>
gap> Dimension(P);
12
```

6.4.2 Vectorize

◇ **Vectorize**(M , $components$) (function)

Returns: a vector in the module M from a list of path algebra elements $components$, which defines the components in the resulting vector.

The returned vector is normalized, so the vector's components may not match the input components.

In the following example, we create two elements in P , perform some elementwise operations, and then construct a submodule using the two elements as generators.

Example

```
gap> p1:=Vectorize(P, [A.b*A.c, A.a*A.c, A.c]);
[ (Z(11)^0)*b*c, (Z(11)^0)*a*c, (Z(11)^0)*c ]
gap> p2:=Vectorize(P, [A.a, A.b, A.w]);
[ (Z(11)^0)*a, (Z(11)^0)*b, (Z(11)^0)*w ]
gap> 2*p1 + p2;
[ (Z(11)^0)*a+(Z(11)^0)*b*c, (Z(11)^0)*b+(Z(11)^0)*a*c, (Z(11)^0)*w+(Z(11)^0)*c ]
```

```
gap> S:=SubAlgebraModule(P,[p1,p2]);
<right-module over <algebra-with-one of dimension 8 over GF(11)>>
gap> Dimension(S);
3
```

6.4.3 \wedge

$\diamond \wedge (m, a)$

(operation)

The operation \wedge operates on an element m in a module and an element a in a path algebra, and it returns the element m multiplied with a . This action is defined by multiplying each component in m by a on the right.

Example

```
gap> p2^(A.c - A.w);
[ (Z(11)^5)*a+(Z(11)^0)*a*c, (Z(11)^5)*b+(Z(11)^0)*b*c,
  (Z(11)^5)*w+(Z(11)^0)*c ]
```

6.4.4 $<$

$\diamond < (m1, m2)$

(operation)

The operation $<$ operates on elements $m1$ and $m2$ in $???$, and it compares them. The result is 'true' if $m1$ is less than $m2$ and false otherwise. Elements are compared componentwise from left to right using the ordering of the underlying algebra. The element $m1$ is less than $m2$ if the first time components are not equal, the component of $m1$ is less than the corresponding component of $m2$.

Example

```
gap> p1 < p2;
false
```

6.4.5 $/$

$\diamond / (M, N)$

(operation)

The operation $/$ operates on two finite dimensional modules M and N over a path algebra?, and it constructs the factor module M/N . This module is again a right algebra module, and all applicable methods and operations are available for the resulting factor module. Furthermore, the resulting module is a vector space, so operations for computing bases and dimensions are also available.

This

Example

```
gap> PS := P/S;
<9-dimensional right-module over <algebra-with-one of dimension
8 over GF(11)>>
gap> Basis(PS);
Basis( <9-dimensional right-module over <algebra-with-one of dimension
8 over GF(11)>>, [ [ [ <zero> of ..., <zero> of ...,
  (Z(11)^0)*w ] ],
  [ [ <zero> of ..., <zero> of ..., (Z(11)^0)*c ] ],
  [ [ <zero> of ..., (Z(11)^0)*v, <zero> of ... ] ],
```

```

[ [ <zero> of ..., (Z(11)^0)*a, <zero> of ... ] ],
[ [ <zero> of ..., (Z(11)^0)*b, <zero> of ... ] ],
[ [ <zero> of ..., (Z(11)^0)*a*c, <zero> of ... ] ],
[ [ <zero> of ..., (Z(11)^0)*b*c, <zero> of ... ] ],
[ [ (Z(11)^0)*v, <zero> of ..., <zero> of ... ] ],
[ [ (Z(11)^0)*b, <zero> of ..., <zero> of ... ] ] ] )

```

Appendix A

An Appendix

This is an appendix.

References

Index

[*](#), [15](#)
[.](#), [12](#), [19](#)
[/](#), [34](#)
[<](#), [16](#), [19](#), [23](#), [34](#)
[=](#), [16](#)
[^](#), [32](#), [34](#)

[AdjacencyMatrixOfQuiver](#), [13](#)
[AdmitsFinitelyManyNontips](#), [27](#)
[ArrowsOfQuiver](#), [12](#)

[Coefficients](#), [23](#)
[CompletelyReduce](#), [25](#)
[CompletelyReduceGroebnerBasis](#), [26](#)

[ElementOfFpPathAlgebra](#), [23](#)
[Enumerator](#), [26](#)

[GeneratorsOfQuiver](#), [13](#)
[GroebnerBasis](#), [24](#)

[IncomingArrowsOfVertex](#), [16](#)
[InDegreeOfVertex](#), [16](#)
[InfoGroebnerBasis](#), [24](#)
[InfoPathAlgebra](#), [18](#)
[InfoQuiver](#), [10](#)
[IsAcyclic](#), [11](#)
[IsArrow](#), [14](#)
[IsCompleteGroebnerBasis](#), [25](#)
[IsCompletelyReducedGroebnerBasis](#), [25](#)
[IsElementOfFpPathAlgebra](#), [22](#)
[IsFinite](#), [12](#)
[IsGroebnerBasis](#), [24](#)
[IsHomogenousGroebnerBasis](#), [25](#)
[IsLeftUniform](#), [20](#)
[IsNormalForm](#), [23](#)
[IsPath](#), [14](#)
[IsPathAlgebra](#), [19](#)
[IsPathAlgebraMatModule](#), [32](#)
[IsPrefixOfTipInTipIdeal](#), [27](#)

[IsQuiver](#), [11](#)
[IsRightGroebnerBasis](#), [28](#)
[IsRightUniform](#), [20](#)
[IsSubAlgebraFpPathAlgebra](#), [22](#)
[IsTipReducedGroebnerBasis](#), [25](#)
[IsUniform](#), [21](#)
[IsVertex](#), [14](#)
[IsZeroPath](#), [14](#)
[Iterator](#), [26](#)

[LeadingCoefficient](#), [20](#)
[LeadingMonomial](#), [20](#)
[LeadingTerm](#), [20](#)
[LengthOfPath](#), [15](#)

[MappedExpression](#), [21](#)

[NakayamaAlgebra](#), [23](#)
[NeighborsOfVertex](#), [17](#)
[Nontips](#), [27](#)
[NontipSize](#), [27](#)
[NormalFormFunction](#), [22](#)

[OrderedBy](#), [11](#), [18](#)
[OrderingOfQuiver](#), [13](#)
[OrderOfQuiver](#), [13](#)
[OutDegreeOfVertex](#), [16](#)
[OutgoingArrowsOfVertex](#), [16](#)

[PathAlgebra](#), [18](#)

[Quiver](#), [10](#)
[QuiverOfPathAlgebra](#), [19](#)

[RightGroebnerBasis](#), [28](#)
[RightGroebnerBasisOfIdeal](#), [28](#)
[RightModuleOverPathAlgebra](#), [30](#)
[RightProjectiveModule](#), [33](#)

[SizeOfQuiver](#), [13](#)
[SourceOfPath](#), [14](#)

TargetOfPath, [15](#)
Tip, [20](#)
TipCoefficient, [20](#)
TipMonomial, [20](#)
TipReduce, [26](#)
TipReduceGroebnerBasis, [26](#)

Vectorize, [33](#)
VerticesOfQuiver, [12](#)

WalkOfPath, [15](#)